МИНОБРНАУКИ РОССИИ

Федеральное государственное автономное образовательное учреждение высшего образования «Южный федеральный университет»

Институт математики, механики и компьютерных наук им. И.И. Воровича

Болотина Анна Сергеевна

ОПРЕДЕЛЕНИЕ ТОЧЕК РЕКУРСИИ В ОБОБЩЁННОМ ПРОГРАММИРОВАНИИ НА ЯЗЫКЕ HASKELL

ВЫПУСКНАЯ КВАЛИФИКАЦИОННАЯ РАБОТА
по направлению 02.03.02 – Фундаментальная информатика и информационные технологии

Научный руководитель –
ст. преп. Брагилевский Виталий Николаевич

Ростов-на-Дону – 2018

Пояснительная записка

Функциональное программирование — это парадигма программирования, при которой программа является выражением, трактуемым как математическая функция, а выполнение программы — это вычисление этого выражения.

Преимущество функциональных программ — в том, что они имеют более простую семантику, точнее описывающую математические объекты, что упрощает доказательство их свойств и корректности. Данная особенность функциональных языков программирования делает их широко применимыми для верификации программного обеспечения. Кроме того, отсутствие побочных эффектов при вычислении выражений делает функциональные программы хорошо поддающимися распараллеливанию. Таким образом, функциональные языки позволяют писать эффективные программы, которые экономят ресурсы и требуют меньше затрат программиста для контроля их качества.

Функциональный стиль программирования предполагает особый взгляд на проблемы и пути размышления над их решением по сравнению с распространёнными языками императивной модели, что формирует образ мышления программиста, использующего функциональные языки, и делает функциональное программирование предметом интереса академических исследований. Идеи из функционального программирования оказывают существенное влияние на развитие языков программирования в целом. Например, такие элементы языков программирования, как лямбда-выражения, сборка мусора и автоматическое управление памятью, поддерживаемые во многих современных языках, заимствованы из функционального программирования.

Другая область в функциональном программировании — обобщённое программирование типов данных (*datatype-generic programming*), предполагающее, особо кодируя данные, описывать

функции, которые можно применять для различных типов. Обобщённое программирование типов данных рассматривает возможность построения некоторого изоморфного представления (*representation*) внутренней структуры для произвольного типа данных и определения функций, параметризованных такими структурными представлениями.

Несколько известных подходов к обобщённому программированию типов данных реализовано в библиотеках на функциональном языке программирования Haskell. Многие широко используемые библиотеки используют различные формы оператора наименьшей неподвижной точки для выражения рекурсивной структуры типов. Другие библиотеки, не использующие неподвижную точку и не кодирующие явно точки рекурсии, позволяют определять обобщённые функции, которые не требуют информации о рекурсивных узлах в структуре типа. Некоторые подходы предлагают использовать небезопасные расширения языка Haskell для работы с рекурсией.

Библиотека обобщённого программирования generics-sop является примером подхода, не кодирующего рекурсивные вхождения в структурном представлении типа. В данной работе рассматривается проблема определения точек рекурсии, вырастающая при использовании данной библиотеки для написания обобщённых функций, траверсирующих структуру типа данных и имеющих дело с рекурсией.

Результатом работы является решение проблемы, использующее замкнутые семейства типов (*closed type families*) — относительно недавнее расширение системы типов Haskell. Полученный метод позволяет описывать функции, требующие информации о точках рекурсии, используя обобщённые представления, не содержащие такой информации. В качестве примера таких функций рассматривается обобщённый интерфейс *зиппера* — структуры данных для эффективной навигации по некоторой исходной древовидной структуре данных.

Тип зиппера, как показывает К. Макбрайд, можно построить на основе обобщённого представления структуры произвольного регу-

лярного типа данных — производный тип соответствует представлению исходной структуры в виде текущего узла, *фокуса*, вместе с окружающим его контекстом. Кроме того, ряд работ демонстрирует, что порождение типа зиппера можно обобщить для систем взаимно рекурсивных типов.

Примеры возможного практического применения зиппера включают реализацию текстового редактора, где точкой фокуса является текущее положение курсора; файловой системы (фокус — рабочий каталог); компилятора; интерактивного средства доказательства теорем. Среди известных примеров использования зиппера в промышленном программном обеспечении — оконный менеджер xmonad и компилятор языка Agda.

Библиотека generics-sop, которая используется в данной работе в качестве учебного примера для апробации полученного решения, распространяется с открытым исходным кодом и доступна в репозитории пакетов Hackage на условиях свободной лицензии BSD-3-Clause.

Данная работа показывает, что обобщённый зиппер для взаимно рекурсивных типов может быть построен в generics-sop с использованием только безопасных расширений языка Haskell, и его реализация не требует специального кодирования рекурсии.

О частичных результатах работы доложено на семинаре «Языки программирования и компиляторы» в Институте математики, механики и компьютерных наук им. И. И. Воровича (Южный федеральный университет). Активные и полезные обсуждения с участниками семинара внесли свой вклад в эту работу.

Доклад о результатах данной работы включён в программу симпозиума «Тенденции в функциональном программировании — 2018», который является ежегодным международным форумом для исследователей, интересующихся всеми аспектами функционального программирования, и предоставляет живую среду для представления последних результатов исследований и идей на будущее.

Contents

Introduction

A classical way to generically express a datatype is to represent its constructors as the chains of nested *binary* sums, and turn constructor arguments into the chains of nested *binary* products [1–3]. De Vries and Löh [6] describe a different sum-of-products approach to representing data using n-ary sums and products that are both lists of types; a sum of products is thus a list of lists of types. They call their view SOP which stands for a "sum of products". It is implemented in the generics-sop [7] library and is based on several relatively recent extensions to the Haskell type system, such as *data kinds*, *kind polymorphism* [8] and *constraint kinds*. Using these Haskell features, the library provides the generic view and equips it with a rich collection of high-level combinators, such as ones for constructing sums and products, collapsing to homogeneous structures, and others. They form an expressive instrument for defining generic functions in a more succinct and high-level style as compared to the classical binary sum-of-products views.

There are many generic functions that deal with the recursive knots when traversing the structure of datatypes. Some of the most general examples are *maps* [9] and *folds* [10]; more advanced one is a *zipper* [11–13]. For handling recursion, several generic programming approaches express datatypes in the form of polynomial functors closed under fixed points [14–16]. The SOP view naturally supports definitions of functions that do not require a knowledge about recursive occurrences, but otherwise becomes unhandy.

One possible solution to the aforementioned shortcoming of SOP is to modify its core by explicitly encoding recursive positions using the fixed-point approach. However, this may complicate the whole framework significantly. Besides, such a decision may lead to extra conversions between the generic views.

Another known solution uses overlapping instances. This, usually unwelcome, Haskell extension complicates reasoning about the seman-

tics of code. In particular, the program behavior becomes unstable, for it can be affected by any module defining more specific instances. Morris and Jones [17] extensively discuss the problems arising from overlapping instances. The overlap problem also strikes in the security setting when code is compiled as `-XSafe` because GHC does not reflect unsafe overlaps and marks the module as safe [18].

We feel both existing approaches unsatisfactory and make the following contributions.

- We describe the problem with the current approach of SOP in detail (Section 1).
- We introduce an idiom that overcomes the problem. The approach avoids both, the use of overlapping instances and changing a generic representation (Section 2).
- We evaluate our approach through the development of a larger-scale use case—the generic zipper. The zipper is meant to be easily and flexibly used with families of mutually recursive datatypes (Section 3).
- We note, that our approach can contribute to the generics-sop's one eliminating some boilerplate instance declarations, which necessarily arise in practice as a consequence of absence of information about recursion points. An example of that, taken from the basic-sop [19] package, is discussed in Section 2.2.

We believe that our idea is suitable for any sum-of-products approach that do not exploit the fixed point view and thus subject to the problem. We choose the generics-sop approach as a case study because it appears to be a widely applicable library and builds on powerful language extensions implemented in GHC.

1. The SOP universe and the problem

In this section, we first review the SOP view on data, describing its basic concepts to introduce the terminology we are using. Then we discuss the problem with handling recursion by generic functions and illustrate it with a short example.

1.1. The SOP view

We first explain the terminology we adopt from SOP [6; 20] and use throughout this thesis. The main idea of the SOP view is to use n-ary sums and products to represent a datatype as an isomorphic code whose kind is a list of lists of types. The SOP approach expresses the code using the DataKinds extension, with a type family:

```
type family Code (a :: *) :: [[*]]
```

An n-ary sum and an n-ary product are therefore modelled as type-level heterogeneous lists: the inner list is an n-ary product that represents a sequence of constructor arguments, while the outer list, an n-ary sum, corresponds to a choice of a particular constructor.

Consider, for instance, a datatype of binary trees:

```
data Tree a = Leaf a | Node (Tree a) (Tree a)
```

This datatype is isomorphic to the following code:

```
type instance Code (Tree a) = '[ '[a], '[Tree a, Tree a]]
```

As shown in Listing 1, the datatypes `NS` for an n-ary sum and `NP` for an n-ary product are defined as GADTs and are *indexed* [12] by a promoted list of types. The encoding also holds an auxiliary type constructor `f` (typically, a functor) which is meant to be applied to every element of the index list. Therefore, `NP` is a modest abstraction over a heterogeneous list.

The definitions of `NS` and `NP` are kind polymorphic. The index list is allowed to contain types of arbitrary kind `k`, since `k` turns to `*` by applying

```
data NP (f :: k → *) (xs :: [k]) where
  Nil  :: NP f '[]
  (:*) :: f x → NP f xs → NP f (x ': xs)
data NS (f :: k → *) (xs :: [k]) where
  Z :: f x → NS f (x ': xs)
  S :: NS f xs → NS f (x ': xs)
```

Listing 1: Datatypes for n-ary sums and products.

the type constructor `f`. Basic instantiations of type parameter `f` found in SOP are identity functor `I`, that is, a type-level equivalent for `id` function, and a constant functor `K`, an analogue of `const`:

```
newtype I (a :: *)          = I {unI :: a}
newtype K (a :: *) (b :: k) = K {unK :: a}
```

If instantiated with `I`, `NP` is a plain heterogeneous list, while `K a` turns it into a homogeneous one, isomorphic to `[a]`. Here is an example value of type `NP I`:

```
I 5 :* I True :* I 'x' :* Nil :: NP I '[Int, Bool, Char]
```

We turn to the sum definition now. The constructor `S` of `NS`, given an index in n-element list, results in an index in a list with $n + 1$ elements, skipping the first one, while `Z` stores the payload of type `f x`. For example, the following chooses the third element of a sum:

```
S (S (Z (I 5))) :: NS I '[Char, Bool, Int, Bool]
```

The sum constructors are similar to Peano numbers, so the choice from a sum of products of a datatype matches the index of its particular constructor in the index list and stores the product representing arguments of that constructor.

With the `NS` and `NP` machinery at hand, SOP defines the `Generic` class with conversion functions `from` and `to` witnessing the isomorphism between a datatype and its generic representation:

```
type Rep a = SOP I (Code a)

class All SListI (Code a) ⇒ Generic (a :: *) where
```

```
type Code a :: [[*]]
from :: a → Rep a
to   :: Rep a → a
```

The sum of products type, `SOP f`, is a newtype-wrapper for `NS (NP f)`, and the structural representation `Rep` of a datatype `a` is a type synonym for a `SOP I` of `a`'s code. The functions, `from` and `to`, perform a shallow conversion of the datatype topmost layer—they do not recursively translate the constructor arguments.

We leave out discussion of the `SListI` constraint in the `Generic` class definition as irrelevant to our work. Although, we do use `All` constraint combinator (as in `All SListI`) in the following. Therefore, it is worth noting that `All` applies a particular constraint (e.g. `SListI` above) to each member of a list of types. The usage of constraints as type arguments is allowed due to ConstraintKinds language extension introducing a dedicated kind `Constraint`.

We have introduced generic representation employed by the SOP library and are ready to describe the problem of handling recursion points, stemming from the representation.

1.2. Problem with handling recursion

We illustrate the problem through a short example. The QuickCheck library [21] for automatic testing of Haskell code defines a helper function `subterms` that takes a term and obtains a list of all its immediate subterms that are of the same type as the given term, that is, all the recursive positions in the term structure. We reimplement this function using the SOP view:

```
subterms :: Generic a ⇒ a → [a]
subterms t = subtermsNS (unSOP $ from t)

subtermsNS :: NS (NP I) xss → [a]
subtermsNS (S ns) = subtermsNS ns
```

```
subtermsNS (Z np) = subtermsNP np

subtermsNP :: ∀a xs.  NP I xs → [a]
subtermsNP p (I y :* ys)
  | typeEq @a y  = witnessEq y :  subtermsNP ys
  | otherwise    = subtermsNP ys
subtermsNP _ Nil = []
```

The function `subterms` translates the term to its representation, unwrapping the sum of products from `SOP`, and passes that to the auxiliary function `subtermsNS`. The latter merely traverses the sum and, once reaches the product, passes it further to `subtermsNP`.

The algorithm of `subtermsNP` is straightforward—it traverses the product, appending current element to the result list if its type is the same as of the term `t`, otherwise skipping the element. We use GHC's TypeApplications extension to pass that type.

Now, we need a way to check type equality and, in the case of equal types, to witness that the element is of the desired type. There is no clear path to this at the moment. Therefore, we step back (until Section 2.1) and, to implement `subtermNP`, follow the QuickCheck's example[1], using overlapping instances of a dedicated class instead.

```
class Subterms a (xs :: [*]) where
  subtermsNP :: NP I xs → [a]

instance Subterms a xs ⇒ Subterms a (x ':  xs) where
  subtermsNP (_   :* xs) = subtermsNP xs
instance {-# OVERLAPS #-}
         Subterms a xs ⇒ Subterms a (a ':  xs) where
  subtermsNP (I x :* xs) = x :  subtermsNP xs
instance Subterms a '[] where
  subtermsNP _ = []
```

[1]The QuickCheck library applies another approach to generic programming, namely GHC.Generics.

To make the whole solution work, we need to propagate the constraints all the way through `subtermsNS` and `subterms` signatures:

```
subterms   :: (Generic a, All (Subterms a) (Code a))
           ⇒ a → [a]
subtermsNS :: All (Subterms a) xss
           ⇒ NS (NP I) xss → [a]
```

Although the approach works, as exemplified by a number of the packages on Hackage, we aim for release of generic programs from overlap. This would remove the complexity overhead introduced by the approach, as we have mentioned in the introduction.

2. Handling recursion with closed type families

In the previous section, we have shown a solution to the problem of handling recursion, which uses overlapping instances. We are going to improve the solution and remove overlap now.

Closed type families are the Haskell language extension introduced by Eisenberg et al. [22]. The main idea of the extension is that the equations for a *closed type family* are disallowed outside its declaration. Under the extension, we can give the following definition of type-level equality:

```
type family Equal a x :: Bool where
  Equal a a = 'True
  Equal a x = 'False
```

The equations in a closed type family are matched in a top-to-bottom order. Since the order is fixed, the overlapping equations here cannot be used to define unsound type-level equations.

2.1. Solution to `subtermsNP` revised

We now return to our running example from Section 1.2. With the type equality, we can witness the coercion between equal types by defining a type class:

```
class Proof (eq :: Bool) (a :: *) (b :: *) where
  witnessEq :: b → Maybe a

instance Proof 'False a b where
  witnessEq _ = Nothing
instance Proof 'True  a a where
  witnessEq   = Just
```

For every element in a list of all direct subterms of a term we shall provide a proof object witnessing its type (in)equality to the type of the term. This can be done by means of the `All` combinator and partially applied auxiliary type class `ProofEq`, which abbreviates the heavy-weighted interface of `Proof`:

```
class    Proof (Equal a b) a b ⇒ ProofEq a b
instance Proof (Equal a b) a b ⇒ ProofEq a b
```

Resulting implementation of `subtermsNP` resembles our first definition given in the previous section:

```
subtermsNP :: ∀a xs.  All (ProofEq a) xs ⇒ NP I xs → [a]
subtermsNP (I (y :: x) :* ys)
  = case witnessEq @(Equal a x) y of
      Just t  → t :  subtermsNP ys
      Nothing → subtermsNP ys
subtermsNP _ Nil = []
```

As a side note, we make use of ScopedTypeVariables extension in the definition above, as the type of the element being matched does not appear in the function signature, since it may match an empty list.

To complete the solution of the problem, the `ProofEq` constraint must be added to the `subterms` and `subtermsNS` declarations as well.

In summary, we claim that any generic function accessing recursive knots in the underlying datatype structure can be defined in the way described above for `subterms` task. We give another example showing how to adapt our idiom to different scenarios in the following subsection.

2.2. Generic show

The function `show` is a common example of useful functions that traverse a datatype's recursive structure. It is known that this function can be defined in a generic way for an arbitrary datatype. De Vries and Löh define generic function `gshow` in the basic-sop package [19] based on the SOP view. We follow their implementation of `gshow` for the most part, but improve it in respect of handling recursion. The original `gshow` yields to the standard `show` generated through **`deriving`** `Show`, because it does not consult with recursion points to place parentheses. We eliminate this drawback.

The following exploits the idea of *pattern matching*. As before, we consider two cases. In the first case, when the position we are matching on is not recursive, we only require it to be an instance of `Show`, and invoke its `show` function. Whereas in the case of the recursive position, we surround it with parentheses and apply our generic function `gshow`. Thus, by means of the type family for equality, we model a form of pattern matching on the types again:

```
class CaseShow (eq :: Bool) (a :: *) (b :: *) where
  caseShow' :: b → String

instance Show  b ⇒ CaseShow 'False a b where
  caseShow'   = show
```

```
instance GShow a ⇒ CaseShow 'True  a a where
  caseShow' t = "(" ++ gshow t ++ ")"
```

We provide a synonym for the `CaseShow (Equal a b) a b` instance, which we call `CaseRecShow`, as before with `ProofEq`; likewise a synonym for the matching function:

```
caseShow :: ∀a b.  CaseRecShow a b ⇒ b → String
caseShow t = caseShow' @(Equal a b) @a t
```

The resulting function `gshow` is a subject of a number of constraints abbreviated by a `GShow` synonym:

```
type GShow a = (Generic a, HasDatatypeInfo a,
                All2 (CaseRecShow a) (Code a))
```

The function `gshow` employs metadata provided by generics-sop's class `HasDatatypeInfo` to show the names of a datatype constructor and its record fields. The generics-sop library is able to derive this metadata automatically. The function is also constrained by `CaseRecShow` with the `All2` combinator that is an analogue of `All` for a list of lists of types.

```
gshow :: ∀a.  GShow a ⇒ a → String
gshow t = gshow' @a (constructorInfo $ datatypeInfo
                                     $ Proxy @a) $ from t
```

The functions from generics-sop use a proxy to fix a type value, where we use TypeApplications—a later language extension than those the library relies on. The auxiliary function `gshow'`, which works on the metadata encoding and generic representation of a datatype, uses generics-sop's combinators for collapsing and mapping:

```
gshow' :: ∀a xss.  All2 (CaseRecShow a) xss
       ⇒ NP ConstructorInfo xss → SOP I xss → String
gshow' cs (SOP sop)
    = hcollapse $ hczipWith allp (goConstructor @a) cs sop
  where
```

```
allp = Proxy @(All (CaseRecShow a))
```

When a sum structure is actually homogeneous (i. e., has type `NS (K a) xs`), it can be collapsed to a single component of type `a`. The respective instantiation of the function `hcollapse` that generalizes collapsing homogeneous structures is

```
hcollapse :: NS (K a) xs → a
```

The `hczipWith` function is generalized `zipWith` that operates with a constrained function on heterogeneous structures, where the proxy fixes the constraint:

```
hczipWith :: All c xs
          ⇒ proxy c → (∀a.  c a ⇒ f a → g a → h a)
          → NP f xs → NS g xs → NS h xs
```

In Listing 2, we show functions that process constructors and record fields

Processing constructors

```
goConstructor :: ∀a xs.  All (CaseRecShow a) xs
              ⇒ ConstructorInfo xs → NP I xs → K String xs
goConstructor (Constructor n) args = K $ unwords (n :  args')
  where
    args' :: [String]
    args' = hcollapse $ hcmap (p @a) (K . caseShow @a .  unI) args
goConstructor (Record n ns) args
    = K $ n ++ " {" ++ intercalate ", " args' ++ "}"
  where
    args' :: [String]
    args' = hcollapse $ hczipWith (p @a) (goField @a) ns args
goConstructor (Infix n _ _) (I arg1 :* I arg2 :* Nil)
    = K $ caseShow @a arg1 ++ " " ++ n ++ " " ++ caseShow @a arg2

p :: Proxy (CaseRecShow a)
p = Proxy
```

Processing record fields

```
goField :: ∀a x.  (CaseRecShow a) x
        ⇒ FieldInfo x → I x → K String x
goField (FieldInfo n) (I y)
    = K $ n ++ " = " ++ caseShow @a y
```

Listing 2: Auxiliary functions for implementing `gshow`.

obtaining their names from proper constructors of metadata and make use of `caseShow`. The function `goConstructor` repeats in general the shape of `gshow'`, where the `hcmap` function generalizes `map` for `NP`, the return type of `hczipWith` on two products becomes `NP`, and on two cases, `hcollapse` collapses the result `NP` to a list of strings.

The function `gshow` can now be used to generically show data—for example, a value of type `Tree Bool`; note that `Tree a` from Section 1.1 is now assumed to be an instance of `Generic` and `HasDatatypeInfo`.

```
*Main> let tree = Node (Leaf True) (Leaf False)
*Main> gshow tree

"Node (Leaf True) (Leaf False)"
```

Here is a benefit of our implementation: it can be used directly, without any additional instance declarations, whereas basic-sop [19] offers the following usage pattern for `gshow` and some datatype `T`:

```
instance Show T where
  show = gshow
```

This is a consequence of `gshow` from basic-sop not treating recursive positions separately, and therefore requiring the `Show` constraint for all knots in the datatype structure.

3. The generic zipper

The zipper is a data structure that enables efficient navigation and editing within the tree-like structure of a datatype. It represents a current location in that structure, storing a tree node, a *focus*, along with its context. Having a zipper focused on a recursive knot in a structure, we may produce a new location by moving the focus up, down, left, or right. On the way, we can update the nodes. Entering and leaving the navigation usually need a special care.

The classical zipper described by Huet [11] can be generically calculated for regular datatypes [12]—a subset of datatypes that can be viewed as a least fixed point of some polynomial expression on types. Yakushev et al. [14] generalize the definition of the generic zipper for an arbitrary family of mutually recursive datatypes. All mentioned solutions require a datatype to be expressed using forms of a fixed-point operator, since the zipper operates on recursion points.

In this section, we describe our approach allowing one to define the generic zipper out of a representation that does not exploit a fixed point. We start with the generic zipper interface and an example of how it can be used (Section 3.1). Then, we turn to the type-level machinery employed to define locations inside mutually recursive datatypes using the SOP view (Section 3.2). Finally, we discuss the implementation of the generic zipper interface — the functions for manipulating locations (Section 3.3).

3.1. Interface and usage

The interface we provide for the generic zipper is shown on Listing 3. It comprises the functions for *movement*, *starting* and *ending navigation*, and *updating* the focus, which are defined over the location structure.

The functions `goUp`, `goDown`, `goLeft`, and `goRight` produce a location with the focus moved *up* to the parent of the focal subtree, *down* to its leftmost child, *left* and *right* to the left and right sibling, respectively, if it is possible. A movement may fail, as specified by the `Maybe` monad, if we cannot go further in a chosen direction. Navigation in a tree starts at the root, and the type variable `a` refers to the root type that remains the same during the navigation, while the type in the focus of the location may vary and is one of the types in a type list `fam`.

The function signature of `enter` specifies the constraints necessary to begin navigation in a structure. Firstly, a datatype of the structure needs to have the `Generic` representation. Secondly, the `In` constraint checks if type `a` is a member of a type family `fam`. Thirdly, the `Zipper` constraint

Movement functions

```
goUp    :: Loc a fam c → Maybe (Loc a fam c)
goDown  :: Loc a fam c → Maybe (Loc a fam c)
goLeft  :: Loc a fam c → Maybe (Loc a fam c)
goRight :: Loc a fam c → Maybe (Loc a fam c)
```

Starting navigation

```
enter   :: ∀fam c a.  (Generic a, In a fam, Zipper a fam c)
        ⇒ a → Loc a fam c
```

Ending navigation

```
leave   :: Loc a fam c → a
```

Updating

```
update  :: (∀b.  c b ⇒ b → b) → Loc a fam c → Loc a fam c
```

Listing 3: Generic zipper interface.

collects specific constraints that refer to the implementation of movement operations. Note that the universal quantifier here sets the instantiation order of the type variables for type applications that will be a part of our usage pattern for the zipper.

The `leave` function ends navigation moving up to the root and returns its modified value.

The `update` function modifies the focal subtree with a given constrained function. The type in focus is existentially quantified inside `Loc` and should satisfy the constraint `c`. The structure of `Loc` (shown in Section 3.2) guarantees that the constraint holds for all types in the family `fam` and, therefore, for all recursive nodes that can be in focus, hence `update` can always be applied.

Consider the following example of usage of the interface. Define a pair of mutually recursive datatypes for a rose tree and a forest, where the forest is a list of trees, and the tree is defined as a value in a node and a forest of its children:

```
data RoseTree a = RTree a (Forest a)

data Forest   a = Empty | Forest (RoseTree a) (Forest a)
```

Updating the trees can be done through a class:

```
class UpdateTree a b where
  replaceBy :: RoseTree a → b → b
  replaceBy _   = id
instance UpdateTree a (RoseTree a) where
  replaceBy t _ = t
instance UpdateTree a (Forest a)
```

This replaces a tree node with a given tree, and, for the forests, this leaves the nodes untouched.

For chaining moves and edits, we can follow Yakushev et al. [14] and employ the flipped function composition >>> and Kleisli composition >=>. The latter is instantiated with the `Maybe` monad that wraps the result type of the movement functions.

```
(>>>) :: (a → b) → (b → c) → (a → c)
(>=>) :: Monad m ⇒ (a → m b) → (b → m c) → (a → m c)
```

The type family we need to run the example is defined as follows:

```
type TreeFam a = '[RoseTree a, Forest a]
```

Finally, we can use zipper operations with our updating function to traverse and replace a part of a forest:

```
*Main> let forest
         = Forest (RTree 'a'
                   $ Forest (RTree 'b' Empty) Empty)
                  (Forest (RTree 'x' Empty) Empty)
*Main> let t = RoseTree 'c' Empty
*Main> enter @(TreeFam Char) @(UpdateTree Char)
         >>> goDown >=> goRight >=> goDown
         >=> update (replaceBy t)
         >>> leave >>> return $ forest
```

This yields the following result:

```
Forest (RTree 'a' $ Forest (RTree 'b' Empty) Empty)
       (Forest (RTree 'c' Empty) Empty)
```

Our zipper applies to regular datatypes as well. In that case, `fam` list shall contain a single element. Generally, the interface is flexible enough to allow us to check in any collection of types we are interested in during traversal. However, we demand an updating operation to be a type class function to distinguish the types of the nodes.

3.2. Locations

The location structure consists of a focal subtree, which is one of the mutually recursive nodes of the whole structure of the family of datatypes, and its surrounding context:

```
data Loc (r :: *) (fam :: [*]) (c :: * → Constraint) where
  Loc :: Focus r a fam c → Contexts r a fam c
      → Loc r fam c
```

The type parameters `r`, `fam`, and `c` in `Loc` correspond to the root type of the tree, the list of types of nodes to visit, and a constraint imposing restrictions on the types in the list, respectively. Also, the single constructor is existentially quantified over one more type variable, `a`, for we need to store a type of the focus' parent to be able to move up successively in a tree-like structure. We discuss both term parameters of the constructor of `Loc` in detail below.

3.2.1. Focus

The subtree in focus is wrapped by the `Focus` datatype. The wrapper incapsulates the proofs about a number of important properties of a focus.

```
data Focus (r :: *) (a :: *) (fam :: [*])
           (c :: * → Constraint) where
```

```
class ProofFocus (inFam :: Bool) (r :: *) (a :: *) (b :: *)
                 (fam :: [*]) (c :: * → Constraint) where
  witness :: b → Maybe (Focus r a fam c)

instance ProofFocus 'False r a b fam c where
  witness _ = Nothing
instance (Generic b, In b fam, ZipperI r a b fam c)
    ⇒ ProofFocus 'True r a b fam c where
  witness   = Just .  Focus

class ProofFocus (InFam b fam) r a b fam c
    ⇒ ProofIn r a b fam c
instance ProofFocus (InFam b fam) r a b fam c
    ⇒ ProofIn r a b fam c
```

Listing 4: Proof of membership of a family of datatypes.

```
Focus :: (Generic b, In b fam, ZipperI r a b fam c)
      ⇒ b → Focus r a fam c
```

Existential type variable `b` represents the type of a focus. We apply a number of predicates to `b`, hence we can implement the steps of the navigation not knowing the actual type of a focus. Firstly, the type of a focus should have the `Generic` representation. Secondly, it should live `In` the list of types we are going to visit. Lastly, it ought to satisfy the set of constraints for the whole zipper interface captured by the `ZipperI` predicate. In particular, the predicate ensures that `a` is the type of the parent for the focus in the structure under consideration.

We implement the `In` constraint by means of a type family `InFam` exactly along the lines of the `Equal` type family defined in the beginning of Section 2.

```
type In a fam = InFam a fam ~ 'True
```

The definition of `InFam` is omitted as a boring one.

One last missing piece for managing focuses is the class `ProofIn`. It provides a proof of membership of a focus type to a family. Again, this generalizes the proof of type equality from Section 2. The definitions of `ProofIn` and an auxiliary class `ProofFocus` are given in Listing 4.

3.2.2. Contexts

A focus on a particular node, augmented with a surrounding context of that node, is enough to reconstruct the entire structure. Therefore, the context of a location has the shape of the original structure but with one hole at the place of its focus. This is sometimes called a *one-hole context*.

The context can be expressed as a stack, called `Contexts`, and each frame, `Context`, corresponds to the particular node with a hole. The stack ascends from the focal node keeping its siblings, the siblings of its parent, etc., until it reaches the root node. So the stack of contexts, essentially, reflects the track of the movement inside the structure.

```
data Contexts (r :: *) (a :: *) (fam :: [*])
              (c :: * → Constraint) where
  CNil :: Contexts a a fam c
  Ctxs :: (Generic a, In a fam, ZipperI r x a fam c)
       ⇒ Context fam a → Contexts r x fam c
       → Contexts r a fam c
```

The type parameters have the same meaning as for the `Loc` datatype. The `ZipperI` constraint with the type `x` of the previous context frame indicates that the constraint for the zipper holds after plugging the focus in the hole. Therefore, all the properties can be proved by induction for the focus type when it moves down in the tree adding new contexts onto the stack. The `CNil` constructor for an empty context, with the `r` and `a` types being equal, forms the inductive basis in that kind of proof.

Note that the type of the current focus is not reflected in the `Contexts` datatype.

3.2.3. Type-level Differentiation

McBride [23] studies a relation between the one-hole context definition and *partial differentiation* from calculus: he shows that the type of the context for an arbitrary (regular) type can be derived mechanically from

that type by means of a list of differentiation *rules* that serve as formulaic instructions for computing the type in type-level programming. Yakushev et al. [14] then demonstrate that the method can be generalized for mutually recursive datatypes. We adapt that technique to generics-sop, and now need a few auxiliary type-level functions to implement the computation of the context type. Those functions, defined recursively via type families, provide algebraic operations for lists of types (which we regard as sums and products of types): addition and multiplication. Specifically, we define addition `.++` of two sums of products (SOP) of types, multiplication `.*` of a SOP by a single type, and multiplication `.**` of a SOP by a product of types, as shown in Listing 5.

SOP addition

```
type family (.++) (xs :: [[*]]) (ys :: [[*]]) :: [[*]] where
  (x ':  xs) .++ ys = x ':  (xs .++ ys)
  '[]       .++ ys = ys
infixr 6 .++
```

SOP-by-type multiplication

```
type family (.*) (x :: *) (ys :: [[*]]) :: [[*]] where
  x .* (ys ':  yss) = (x ':  ys) ':  (x .* yss)
  x .* '[]         = '[]
infixr 7 .*
```

SOP-by-product multiplication

```
type family (.**) (xs :: [*]) (ys :: [[*]]) :: [[*]] where
  (x ':  xs) .** yss = x .* (xs .** yss)
  '[]       .** yss = yss
infixr 7 .**
```

Listing 5: Algebraic operations on type-level sums and products.

The addition operation just appends two type-level lists of lists (sums of products), multiplication by a type adds the type to the head of each inner list of the sum (here we see multiplication of a product and the distributive property of multiplication over addition, just as in arithmetic of numbers), and multiplication by a product appends the list to the head of each inner product of the sum. Again, kind `[*]` here denotes products, and `[[*]]` denotes sums (of products), so the relation with arithmetic of

numbers in these definitions becomes more clear if one realizes that an empty sum `'[] :: [[*]]` corresponds to 1, and an empty product `'[] ::: [*]` corresponds to 0. We also specify, through the `infixr` declaration, that multiplication has a higher priority than addition.

3.2.4. Context Frame

At this point, we can implement differentiation of a product of types and, therefore, the computation of a context frame type.

The definition of differentiation, shown in Listing 6, resembles its analogue from calculus, but it is now generalized for the setting of families of datatypes: the differentiation of the single type reflected by a one-element list here results in 0 (reflected by `'[]`) if that is not in the family and hence is regarded as a constant, otherwise it results in 1. When differentiation gives 1, it is actually the hole, which we express by defining the unit type `Hole`. Reflecting this type in the context is helpful when we traverse the context representation to plug the hole. We use the empty type `End` to distinguish the case when the hole is found at the end of the list, in order not to add that into the result twice. The sum `'[ '[]]` represents a unit type that is exactly 1. We also use type-level `If` that returns its second argument for `'True`, and the third one otherwise. We do not give its definition here, as it is straightforward.

The following completes the computation of the context type:

```
data ConsN = F | N ConsN | None


data Hole = Hole
data End
type family DiffProd (fam :: [*]) (xs :: [*]) :: [[*]] where
  DiffProd fam '[]       = '[]
  DiffProd fam '[x]      = If (InFam x fam) '[ '[Hole]] '[]
  DiffProd fam '[End, x] = If (InFam x fam) '[ '[]]     '[]
  DiffProd fam (x ':  xs)
    = Hole .* xs .** DiffProd fam '[End, x] .++
              x  .*  DiffProd fam xs
```

Listing 6: Differentiation of a product of types.

```
  deriving Eq

type family ToContext (n :: ConsN) (fam :: [*])
                      (code :: [[*]]) :: [[*]] where
  ToContext n fam '[] = '[]
  ToContext n fam (xs ':  xss)
    = Proxy n .* DiffProd fam xs
              .++ ToContext ('N n) fam xss
```

The type family `ToContext` derives the type of the context of a datatype performing differentiation of a sum on its code. Since each product from the code matches a sum of multiple products in the context, it is helpful for each product of the context representation, to keep the index of its matching constructor of the datatype. We store the index in the datatype `ConsN` adding that to the head of each product but wrapped by `Proxy` because we use `ConsN` promoted to a kind in the type family, while the products contain types of kind `*`. The constructors `F` and `N` denote "first" and "next", respectively, and the special constructor `None` will be used further to indicate failure of matching indices.

Finally, a context frame has two representations: as a type synonym and as a newtype wrapper.

```
type CtxCode fam a = ToContext 'F fam (Code a)

newtype Context fam a = Ctx {ctx :: SOP I (CtxCode fam a)}
```

The wrapper allows GHC to perform type inference where it is doomed to fail with a plain type synonym. On the other hand, the `CtxCode` type comes handy in constraints applied to the generic zipper interface functions.

3.3. Implementing the zipper interface

We now can implement the interface functions of the zipper, which we have previously described. We only demonstrate the implementation of the `goDown` function here. This shows the idea of how we can use our

```
toFirst :: ∀fam c r a.  (Generic a, ToFirst r a fam c)
        ⇒ a → Maybe (Focus r a fam c)
toFirst t = appToNP @AllProof toFirstNP $ unSOP $ from t
```

Proof

```
class    All (ProofIn r a fam c) xs ⇒ AllProof r a fam c xs
instance All (ProofIn r a fam c) xs ⇒ AllProof r a fam c xs

type ToFirst r a fam c = All (AllProof r a fam c) (Code a)
```

Processing products

```
toFirstNP :: ∀fam c r a xs.  All (ProofIn r a fam c) xs
          ⇒ NP I xs → Maybe (Focus r a fam c)
toFirstNP (I (x :: b) :* xs)
  = witness @(InFam b fam) x `mplus` toFirstNP xs
toFirstNP Nil = Nothing
```

Listing 7: Implementation of `toFirst`.

idiom for defining the zipper functions, and the source code with the full implementation of the zipper interface is available at our GitHub repository[2].

To move focus down to the leftmost child of the current focal node in the tree, we should analyze the focal subtree's representation to find its first immediate child, and compute its respective context. The following definition of `goDown` uses two auxiliary functions: `toFirst` and `toFirstCtx`.

```
goDown :: Loc a fam c → Maybe (Loc a fam c)
goDown (Loc (Focus t) cs)
  = case toFirst t of
      Just t' → Just $ Loc t' (Ctxs (toFirstCtx t) cs)
      _       → Nothing
```

The function `toFirst` returns its result in the `Maybe` monad, and may return `Nothing`, if the focal node has no children, that is, we currently focus on the leaf node and cannot go down. The function `toFirstCtx` should not fail: if we can move, it computes the context that matches the leftmost subtree selected from the focus' children.

[2] `https://github.com/Maryann13/Zipper`

3.3.1. `toFirst`

We first implement the function `toFirst`. Its full definition is displayed in Listing 7. The `toFirst` function uses the higher-oredered function `appToNP` that unwraps the product from `NS` and applies the given function to that product. The function `toFirstNP` is defined recursively using the proof we have defined for families: it traverses the product until it finds the first recursive node by means of `witness` that for the node `x` of unknown type `b`, witnesses its membership of the family, or else returns `Nothing`. To provide the proof for the representation code of a datatype, we define the proof for all products in a sum, and pass this proof through explicit type application to `appToNP` which takes a constrained function. The `appToNP` function is defined similarly to `subtermsNS` from Section 1.2, and we omit its definition here.

3.3.2. `toFirstCtx`

The definition of `toFirstCtx` is more complicated, as it performs the computation of the context. The implementation comprises several steps including type- and term-level programming. In the following, we systematically construct the context representation from the given generic representation of a datatype.

At first, for a datatype's given constructor represented by `NP`, we build its matching constructor of the context. All constructors of the context have the same shape as the constructors of its respective datatype but with a hole at one of points of recursion—we are now computing the context with the first recursive node deleted. Assuming that we have the product type for the context computed, we can compute the product by matching on that type:

```
class FromFstRec (ys :: [*]) (xs :: [*]) where
  fromFstRec :: NP I xs → NP I ys

instance FromFstRec (Hole ': xs) (x ': xs) where
```

```
  fromFstRec (_ :* xs) = I Hole :* xs
instance FromFstRec ys xs
    ⇒ FromFstRec (x ': ys) (x ': xs) where
  fromFstRec (x :* xs) = x      :* fromFstRec xs
instance FromFstRec '[] '[] where
  fromFstRec _         = Nil
```

Note that the first type parameter in the `FromFstRec` class is the index type list of the result `NP`—this order of type variables remains for type application through which we will supply the computed type.

Once we have constructed the product, we have to build the sum representing the choice of that product. If we have the index of the chosen constructor for the datatype, the one that we find for its context can be computed as follows:

```
type family CtxConsN (xss :: [[*]])
                     (n :: ConsN) :: ConsN where
  CtxConsN '[]                          n = 'None
  CtxConsN ((Proxy n  ': xs) ': xss) n = 'F
  CtxConsN ((Proxy n' ': xs) ': xss) n
    = 'N (CtxConsN xss n)
```

The list `xss` here is expected to be the context code, which this traverses until the first product storing the constructor index equal to the given one, i. e., the context's first constructor matching with the chosen constructor of the datatype.

To construct the sum with the computed index and product, we again need a proof to witness that the product is type-consistent with the constructor choice. To choose the constructor from the context code, we can adopt generics-sop's *injections*[3]:

```
injections :: ∀xs f.  SListI xs ⇒ NP (Inj f xs) xs
```

[3]The actual definition of the type of injections in generics-sop slightly differs from this. We adapt that for presentation.

```
newtype Inj f xs a = Inj {apInj :: f a → K (NS f xs) a}
```

For `f` instantiated to `NP I` and `xs` to be the sum of products reflecting a representation code, `injections` creates a product of all constructor choices that inject appropriate constructor arguments into each sum. We can use injections to choose the one that matches the obtained index.

We now can witness the choice using the proof of type equality we have defined in Section 2. In the following definition, `f` is supposed to be instantiated to `Inj (NP I) xss` where `xss` reflects the code of the context[4]:

```
class    Proof (Equal a b) (f a) (f b) ⇒ ProofF f a b
instance Proof (Equal a b) (f a) (f b) ⇒ ProofF f a b

class ConsNInj (n :: ConsN) (ys :: [*]) where
  consNInj :: All (ProofF f ys) xss ⇒ NP f xss → f ys

instance ConsNInj n xs ⇒ ConsNInj ('N n) xs where
  consNInj (_ :* xss) = consNInj @n xss
  consNInj Nil        = impossible
instance ConsNInj 'F xs where
  consNInj ((xs :: f xs) :* _)
    = fromMaybe impossible $ witnessEq @(Equal xs ys) xs
  consNInj Nil = impossible
instance ConsNInj 'None xs where
  consNInj _   = impossible

impossible :: a
impossible = error "impossible"
```

As long as the constructor index and product type are computed correctly, the proof should never fail (`impossible`). And when it passes, this ensures by type check that the constructor choice for the given product type is faithful.

[4] This might be simplified by using a singleton type for `ConsN` instead of defining a type class function. However, when using a singleton, the recursive definition of `toFirstCtx` we give further requires a constraint on its function signature that leads to a nonterminating computation.

Finally, we define a function that applies the injection and returns the constructed context for the given product, representing its constructor arguments, and index of that constructor. The type of `injections` below will be inferred from the return type of the context.

```
type AppInj n xs ctx
    = (ConsNInj n xs, SListI ctx,
       All (ProofF (Inj (NP I) ctx) xs) ctx)

appInjCtx :: ∀n xs fam a.  AppInj n xs (CtxCode fam a)
          ⇒ NP I xs → Context fam a
appInjCtx np
  = Ctx $ SOP $ unK $ apInj (consNInj @n injections) np
```

The definition of `toFirstCtx` can now be given using the defined functions `appInjCtx` and `fromFstRec`. The following makes use of a type-level function `FstRecToHole` to compute the product type by replacing the first recursive occurrence in the given datatype's product list with type `Hole`. Its definition is straightforward, and is omitted here.

```
toFirstCtx :: ∀fam c a.  (Generic a, ToFirstCtx fam a)
           ⇒ a → Context fam a
toFirstCtx t = toFirstCtxNS @'F $ unSOP $ from t

toFirstCtxNS :: ∀n fam a xss.  ToFirstCtx' fam a n xss
             ⇒ NS (NP I) xss → Context fam a
toFirstCtxNS (S ns) = toFirstCtxNS @('N n) ns
toFirstCtxNS (Z (np :: NP I xs))
  = appInjCtx @(CtxConsN (CtxCode fam a) n) $
        I (Proxy @n) :* fromFstRec @(FstRecToHole fam xs) np


type ToFirstCtx fam a = ToFirstCtx' fam a 'F (Code a)
```

Here, `ToFirstCtx'` is a complex constraint for use of `appInjCtx` and `fromFstRec`, which involves extra type classes and a bit of type-level

programming to deduce those particular constraints. Its definition is shown in Listing 8.

```
type FstCtx n fam xs = Proxy n ':  FstRecToHole fam xs

class ConsNInj (CtxConsN (CtxCode fam a) n) (FstCtx n fam xs)
  ⇒ CtxConsNInj fam a xs n
instance ConsNInj (CtxConsN (CtxCode fam a) n) (FstCtx n fam xs)
  ⇒ CtxConsNInj fam a xs n

class All (ProofF f (FstCtx n fam xs)) yss
  ⇒ CtxAllProofF f fam yss xs n
instance All (ProofF f (FstCtx n fam xs)) yss
  ⇒ CtxAllProofF f fam yss xs n

class FromFstRec (FstRecToHole fam xs) xs
  ⇒ FromFstRec' fam xs
instance FromFstRec (FstRecToHole fam xs) xs
  ⇒ FromFstRec' fam xs

type family AllNum (c :: k → ConsN → Constraint)
                   (xs :: [k]) (n :: ConsN) :: Constraint where
  AllNum c '[]       n = ()
  AllNum c (x ':  xs) n = (c x n, AllNum c xs ('N n))

type ToFirstCtx' fam a n xss
    = (SListI (CtxCode fam a),
       All (FromFstRec' fam) xss,
       AllNum (CtxConsNInj fam a) xss n,
       AllNum (CtxAllProofF
                 (Inj (NP I) (CtxCode fam a))
                 fam (CtxCode fam a)) xss n)
```

Listing 8: Definition of `ToFirstCxt'`.

We have established the mechanism of translation, which is cumbersome but verifying, via the proof, that the constructed context will have the correct type for any given datatype and family. The other functions in the zipper interface can be implemented according to this technique.

4. Related work

There are many works that contribute to the datatype-generic programming. Rodriguez et al. [4] and Magalhães and Löh [24] review a number of existing approaches and provide their detailed comparison in vari-

ous aspects. There are several generic views that use certain forms of the fixed point operator to express recursion in a datatype structure [1; 14–16]. And there are a number of approaches that do not make use of fixed points [2; 9; 25; 26], but explicitly encode recursion in the datatype representation. The SOP view [6], which we use to demonstrate our technique, is an approach to generic programming that does not reflect recursive positions in the generic representation of a datatype. This approach uses heterogeneous lists of types to encode sums and products in the generic representation.

The idea similar to SOP has been proposed by Kiselyov et al. [27] in their HList library for strongly typed heterogeneous collections. In the paper, the authors also discuss problems connected with overlap, which they use for access operations. Another Haskell extension, functional dependencies, is applied to restrict overlap by introducing a class for type equality there, which resembles our solution.

Morris and Jones [17] introduce the type-class system ilab, based on the Haskell 98 class system, with a new feature called “instance chains”. This enables one to control overlap by using an explicit syntax in instance declarations. The approach resembles if-else chains. But the use of instance chains and local use of overlap leave code error-prone as a consequence of type class openness. Closed type families [22] were recently introduced in Haskell to solve the overlap problem.

Several works show how to define the Zipper [11] generically for regular [12; 23] and mutually recursive [14] types using fixed-point generic views. Adams [13] defines a generic zipper for heterogeneous types.

Conclusion

Defining generic functions, which consider recursion points, is easy within generic views that are explicit about recursion in the datatype representation. Not so much otherwise. Although, there are some approaches

that address the problem by means of global or local overlaps. We have developed the technique that allows one to define generic functions that treat recursion without its explicit encoding and without overlap.

We have demonstrated that the method suits for advanced recursive schemes, such as the generic zipper interface. Also, it supports families of mutually recursive datatypes.

Arguably, it is still easier to treat recursion when "explicit" encoding is used. On the other hand, we believe, once the problem of handling recursion is shown to be manageable, new generic universes shall emerge, not worrying about the recursion support, but rather focusing on other generic programming problems.

Bibliography

1. A Lightweight Approach to Datatype-generic Rewriting / T. Van Noort [et al.] // Proceedings of the ACM SIGPLAN Workshop on Generic Programming. — Victoria, BC, Canada : ACM, 2008. — Pp. 13–24. — (WGP '08). — ISBN 978-1-60558-060-9. — DOI: 10.1145/1411318.1411321.
2. *Cheney J.*, *Hinze R.* A Lightweight Implementation of Generics and Dynamics // Proceedings of the 2002 ACM SIGPLAN Workshop on Haskell. — Pittsburgh, Pennsylvania : ACM, 2002. — Pp. 90–104. — (Haskell '02). — ISBN 1-58113-605-6. — DOI: 10.1145/581690.581698.
3. *Löh A.* Exploring Generic Haskell: PhD thesis / Löh Andres. — Utrecht University, 2004.
4. Comparing Libraries for Generic Programming in Haskell / A. Rodriguez [et al.] // Proceedings of the First ACM SIGPLAN Symposium on Haskell. — Victoria, BC, Canada : ACM, 2008. — Pp. 111–

122. — (Haskell '08). — ISBN 978-1-60558-064-7. — DOI: 10.1145/1411286.1411301.

5. *Magalhães J. P.* Less Is More: Generic Programming Theory and Practice: PhD thesis / Magalhães José Pedro. — Utrecht University, 2012.

6. *De Vries E.*, *Löh A.* True Sums of Products // Proceedings of the 10th ACM SIGPLAN Workshop on Generic Programming. — Gothenburg, Sweden : ACM, 2014. — Pp. 83–94. — (WGP '14). — ISBN 978-1-4503-3042-8. — DOI: 10.1145/2633628.2633634.

7. *De Vries E.*, *Löh A.* generics-sop: Generic Programming using True Sums of Products. — 2018. — URL: http://hackage.haskell.org/package/generics-sop.

8. Giving Haskell a Promotion / B. A. Yorgey [et al.] // Proceedings of the 8th ACM SIGPLAN Workshop on Types in Language Design and Implementation. — Philadelphia, Pennsylvania, USA : ACM, 2012. — Pp. 53–66. — (TLDI '12). — ISBN 978-1-4503-1120-5. — DOI: 10.1145/2103786.2103795.

9. A Generic Deriving Mechanism for Haskell / J. P. Magalhães [et al.] // Proceedings of the Third ACM Haskell Symposium on Haskell. — Baltimore, Maryland, USA : ACM, 2010. — Pp. 37–48. — (Haskell '10). — ISBN 978-1-4503-0252-4. — DOI: 10.1145/1863523.1863529.

10. *Meijer E.*, *Fokkinga M.*, *Paterson R.* Functional programming with bananas, lenses, envelopes and barbed wire // Functional Programming Languages and Computer Architecture / ed. by J. Hughes. — Berlin, Heidelberg : Springer Berlin Heidelberg, 1991. — Pp. 124–144. — ISBN 978-3-540-47599-6.

11. *Huet G.* The Zipper // Journal of Functional Programming. — 1997. — Vol. 7, no. 5. — Pp. 549–554.

12. *Hinze R.*, *Jeuring J.*, *Löh A.* Type-indexed data types // Science of Computer Programming. — 2004. — Vol. 51, no. 1. — Pp. 117–151. — ISSN 0167-6423. — DOI: 10.1016/j.scico.2003.07.001. — Mathematics of Program Construction (MPC 2002).

13. *Adams M. D.* Scrap Your Zippers: A Generic Zipper for Heterogeneous Types // Proceedings of the 6th ACM SIGPLAN Workshop on Generic Programming. — Baltimore, Maryland, USA : ACM, 2010. — Pp. 13–24. — (WGP '10). — ISBN 978-1-4503-0251-7. — DOI: 10.1145/1863495.1863499.

14. Generic Programming with Fixed Points for Mutually Recursive Datatypes / A. R. Yakushev [et al.] // Proceedings of the 14th ACM SIGPLAN International Conference on Functional Programming. — Edinburgh, Scotland : ACM, 2009. — Pp. 233–244. — (ICFP '09). — ISBN 978-1-60558-332-7. — DOI: 10.1145/1596550.1596585.

15. *Jansson P.*, *Jeuring J.* PolyP—a Polytypic Programming Language Extension // Proceedings of the 24th ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages. — Paris, France : ACM, 1997. — Pp. 470–482. — (POPL '97). — ISBN 0-89791-853-3. — DOI: 10.1145/263699.263763.

16. *Löh A.*, *Magalhães J. P.* Generic Programming with Indexed Functors // Proceedings of the Seventh ACM SIGPLAN Workshop on Generic Programming. — Tokyo, Japan : ACM, 2011. — Pp. 1–12. — (WGP '11). — ISBN 978-1-4503-0861-8. — DOI: 10.1145/2036918.2036920.

17. *Morris J. G.*, *Jones M. P.* Instance Chains: Type Class Programming Without Overlapping Instances // Proceedings of the 15th ACM SIGPLAN International Conference on Functional Programming. — Baltimore, Maryland, USA : ACM, 2010. — Pp. 375–386. — (ICFP '10). — ISBN 978-1-60558-794-3. — DOI: 10.1145/1863543.1863596.

18. Safe Haskell & Overlapping Instances—GHC. — 2015. — URL: https://ghc.haskell.org/trac/ghc/wiki/SafeHaskell/NewOverlappingInstances.

19. basic-sop: Basic examples and functions for generics-sop. — 2017. — URL: https://hackage.haskell.org/package/basic-sop.

20. *Löh A.* Applying Type-Level and Generic Programming in Haskell. — 2018. — URL: https://github.com/kosmikus/SSGEP/blob/master/LectureNotes.pdf ; Summer School on Generic and Effectful Programming (SSGEP 2015).

21. *Claessen K.*, *Hughes J.* QuickCheck: A Lightweight Tool for Random Testing of Haskell Programs // SIGPLAN Not. — New York, NY, USA, 2011. — May. — Vol. 46, no. 4. — Pp. 53–64. — ISSN 0362-1340. — DOI: 10.1145/1988042.1988046.

22. Closed Type Families with Overlapping Equations / R. A. Eisenberg [et al.] // Proceedings of the 41st ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages. — San Diego, California, USA : ACM, 2014. — Pp. 671–683. — (POPL '14). — ISBN 978-1-4503-2544-8. — DOI: 10.1145/2535838.2535856.

23. *McBride C.* The Derivative of a Regular Type is its Type of One-Hole Contexts. — 2001. — URL: http://strictlypositive.org/diff.pdf ; Unpublished manuscript.

24. *Magalhães J. P.*, *Löh A.* A Formal Comparison of Approaches to Datatype-Generic Programming // Proceedings Fourth Workshop on Mathematically Structured Functional Programming, MSFP@ETAPS 2012, Tallinn, Estonia, 25 March 2012. — 2012. — Pp. 50–67. — DOI: 10.4204/EPTCS.76.6.

25. *Chakravarty M. M. T.*, *Ditu G. C.*, *Leshchinskiy R.* Instant Generics: Fast and Easy. — 2009. — URL: http://www.cse.unsw.edu.au/~chak/papers/CDL09.html.

26. *Weirich S.* RepLib: A Library for Derivable Type Classes // Proceedings of the 2006 ACM SIGPLAN Workshop on Haskell. — Portland, Oregon, USA : ACM, 2006. — Pp. 1–12. — (Haskell '06). — ISBN 1-59593-489-8. — DOI: 10.1145/1159842.1159844. — URL: http://doi.acm.org/10.1145/1159842.1159844.

27. *Kiselyov O.*, *Lämmel R.*, *Schupke K.* Strongly Typed Heterogeneous Collections // Proceedings of the 2004 ACM SIGPLAN Workshop on Haskell. — Snowbird, Utah, USA : ACM, 2004. — Pp. 96–107. — (Haskell '04). — ISBN 1-58113-850-4. — DOI: 10.1145/1017472.1017488.